



A1	BUREAUTIQUE	
	TP18 Les événements	

**Objectif :** Utiliser les événements en VBA

## LES ÉVÉNEMENTS


Jusque-là, nous avons lancé nos macros en cliquant sur un bouton.

Il est également possible de les exécuter automatiquement lors d'un événement particulier du classeur, tel que l'ouverture du classeur, sa fermeture, son enregistrement, etc.

### A. Workbook

## WORKBOOK\_OPEN (À L'OUVERTURE)

Pour exécuter des instructions à l'ouverture du classeur (après l'activation des macros par l'utilisateur), rendez-vous dans ThisWorkbook et sélectionnez Workbook :



L'événement Workbook\_Open est ajouté par défaut, il agit à l'ouverture du classeur :


```
Private Sub Workbook_Open()
End Sub
```

Par exemple, en ajoutant l'instruction suivante, une boîte de dialogue sera affichée à l'ouverture du classeur :

```
Private Sub Workbook_Open()
    MsgBox "Message de bienvenue"
End Sub
```

## WORKBOOK\_BEFORECLOSE (AVANT FERMETURE)

Pour exécuter des instructions juste avant la fermeture du classeur, choisissez BeforeClose :



```
Private Sub Workbook_BeforeClose(Cancel As Boolean)
```

```
End Sub
```

La fermeture du classeur peut être annulée en attribuant la valeur True à la variable Cancel.

Voici un exemple où l'utilisateur doit confirmer la fermeture du classeur :

```
Private Sub Workbook_BeforeClose(Cancel As Boolean)

    'Si l'utilisateur répond Non, la variable Cancel vaudra True (ce qui annulera la
    fermeture)
    If MsgBox("Etes-vous certain de vouloir fermer ce classeur ?", 36, "Confirmation")
= vbNo Then
        Cancel = True
    End If
End Sub
```

## WORKBOOK\_BEFORESAVE (AVANT ENREGISTREMENT)

Cet événement se déclenche juste avant l'enregistrement :

```
Private Sub Workbook_BeforeSave(ByVal SaveAsUI As Boolean, Cancel As Boolean)

End Sub
```

L'enregistrement peut être annulé en attribuant la valeur True à la variable Cancel.

## WORKBOOK\_AFTERSAVE (APRÈS ENREGISTREMENT)

Cet événement se déclenche juste après l'enregistrement :

```
Private Sub Workbook_AfterSave(ByVal Success As Boolean)

End Sub
```

## WORKBOOK\_BEFOREPRINT (AVANT IMPRESSION)

Cet événement se déclenche juste avant l'impression :

```
Private Sub Workbook_BeforePrint(Cancel As Boolean)

End Sub
```

L'impression peut être annulée en attribuant la valeur True à la variable Cancel.

## WORKBOOK\_SHEETACTIVATE (À L'ACTIVATION D'UNE FEUILLE)

Cet événement se déclenche à chaque changement de feuille :

```
Private Sub Workbook_SheetActivate(ByVal Sh As Object)

End Sub
```

Par exemple, affichage du nom de la feuille dans une boîte de dialogue :

```
Private Sub Workbook_SheetActivate(ByVal Sh As Object)

    MsgBox "Nom de la feuille : " & Sh.Name

End Sub
```

## WORKBOOK\_SHEETBEFOREDOUBLECLICK (AVANT DOUBLE-CLIC)

Cet événement se déclenche juste avant un double-clic sur une cellule :

```
Private Sub Workbook_SheetBeforeDoubleClick(ByVal Sh As Object, ByVal Target As Range, Cancel As Boolean)

End Sub
```

Par exemple, coloration d'une cellule double-cliquée en fonction de la feuille :

```
Private Sub Workbook_SheetBeforeDoubleClick(ByVal Sh As Object, ByVal Target As Range, Cancel As Boolean)

    If Sh.Name = "Feuill1" Then
        Target.Interior.Color = RGB(255, 108, 0) 'Couleur orange
    Else
        Target.Interior.Color = RGB(136, 255, 0) 'Couleur verte
    End If

End Sub
```

Le double-clic peut être annulé en attribuant la valeur True à la variable Cancel.

## WORKBOOK\_SHEETBEFORERIGHTCLICK (AVANT CLIC DROIT)

Cet événement se déclenche juste avant un clic droit sur une cellule :

```
Private Sub Workbook_SheetBeforeRightClick(ByVal Sh As Object, ByVal Target As Range, Cancel As Boolean)

End Sub
```

## WORKBOOK\_SHEETCHANGE (À CHAQUE MODIFICATION DE CELLULE)

Cet événement se déclenche à chaque modification du contenu d'une cellule :

```
Private Sub Workbook_SheetChange(ByVal Sh As Object, ByVal Target As Range)

End Sub
```

## WORKBOOK\_SHEETSELECTIONCHANGE (À CHAQUE CHANGEMENT DE SÉLECTION)

Cet événement se déclenche à chaque changement de sélection sur une feuille de calcul :

```
Private Sub Workbook_SheetSelectionChange(ByVal Sh As Object, ByVal Target As Range)

End Sub
```

Par exemple, coloration de la sélection si A1 est vide :

```
Private Sub Workbook_SheetSelectionChange(ByVal Sh As Object, ByVal Target As Range)

    If Range("A1") = "" Then
        Target.Interior.Color = RGB(124, 255, 255) 'Bleu clair
    End If

End Sub
```

## WORKBOOK\_NEWSHEET (À L'INSERTION D'UNE FEUILLE)

Cet événement se déclenche à chaque insertion d'une nouvelle feuille :

```
Private Sub Workbook_NewSheet(ByVal Sh As Object)


End Sub
```

A la page précédente les événements concernaient le classeur entier. Sur celle-ci, nous allons nous focaliser sur les événements liés à une feuille.

## B. Worksheet

### WORKSHEET\_SELECTIONCHANGE (AU CHANGEMENT DE SÉLECTION)

Pour exécuter des instructions en fonction d'un événement pour une feuille en particulier, sélectionnez la feuille dans l'éditeur, puis Worksheet :



L'événement SelectionChange est ajouté par défaut, il agit lors d'un changement de sélection :

```
Private Sub Worksheet_SelectionChange(ByVal Target As Range)

End Sub
```

Par exemple, voici un code qui colore la ou les cellules sélectionnées et qui supprime automatiquement la coloration de la dernière sélection lors d'un changement de sélection :

```
Private Sub Worksheet_SelectionChange(ByVal Target As Range)

    Static selectionPrecedente As String

    'Suppression de la couleur de fond de la sélection précédente
    If selectionPrecedente <> "" Then
        Range(selectionPrecedente).Interior.ColorIndex = xlColorIndexNone
    End If

    'Coloration de la sélection actuelle
    Target.Interior.Color = RGB(181, 244, 0)

    'Enregistrement de l'adresse de la sélection actuelle
    selectionPrecedente = Target.Address

End Sub
```

### WORKSHEET\_ACTIVATE (À L'ACTIVATION DE LA FEUILLE)

Cet événement se déclenche lorsque la feuille est activée :

```
Private Sub Worksheet_Activate()

End Sub
```

Par exemple, sélection de la cellule D5 à l'activation de la feuille :

```
Private Sub Worksheet_Activate()  
    Range("D5").Select  
End Sub
```

## WORKSHEET\_DEACTIVATE (À LA SORTIE DE LA FEUILLE)

Cet événement se déclenche lorsqu'une autre feuille du classeur est activée :

```
Private Sub Worksheet_Deactivate()  
  
End Sub
```

Par exemple, effacement du contenu des cellules B2 à B10 après avoir quitté la feuille :

```
Private Sub Worksheet_Deactivate()  
    Range("B2:B10").ClearContents  
End Sub
```

## WORKSHEET\_BEFOREDOUBLECLICK (AU DOUBLE-CLIC)

Cet événement se déclenche lors d'un double-clic sur une cellule de la feuille :

```
Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, Cancel As Boolean)  
  
End Sub
```

Par exemple, coloration de la cellule double-cliquée en vert (ou en blanc si elle est déjà colorée) :

```
Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, Cancel As Boolean)  
    If Target.Interior.Color = 16777215 Then 'Si blanc  
        Target.Interior.Color = RGB(200, 255, 100) 'Couleur verte  
    Else 'Sinon  
        Target.Interior.Color = 16777215 'Couleur blanche  
    End If  
End Sub
```

Le double-clic peut être annulé en attribuant la valeur True à la variable Cancel.

## WORKSHEET\_BEFORERIGHTCLICK (AU CLIC DROIT)

Cet événement se déclenche lors d'un clic droit sur la feuille :

```
Private Sub Worksheet_BeforeRightClick(ByVal Target As Range, Cancel As Boolean)  
  
End Sub
```

Par exemple, ajout de la **date du jour** par clic droit si la cellule cliquée est dans la colonne C :

```
Private Sub Worksheet_BeforeRightClick(ByVal Target As Range, Cancel As Boolean)

    'Si cellule de la colonne 3 (C)
    If Target.Column = 3 Then
        Target = Date 'Ajout de la date du jour
        Cancel = True 'Annulation du clic droit
    End If

End Sub
```

Le clic droit peut être annulé en attribuant la valeur True à la variable Cancel (dans ce cas, le menu contextuel ne sera pas affiché).

## WORKSHEET\_CHANGE (À CHAQUE MODIFICATION DE CELLULE)

Cet événement se déclenche lors de modifications du contenu des cellules de la feuille :

```
Private Sub Worksheet_Change(ByVal Target As Range)

End Sub
```

## DÉSACTIVER TEMPORAIREMENT TOUS LES ÉVÉNEMENTS

Pour exécuter du code sans déclencher d'événements, placez-le entre ces deux lignes :

```
Application.EnableEvents = False 'Désactive les événements

'Instructions ...

Application.EnableEvents = True 'Réactive les événements
```